*==========================================================================================================================================================
*
*	Local Projections
*	Felix Ward
*
*==========================================================================================================================================================


set more off

qui {

*-------------------------------------------------------------------------------------------------------------
*	SETTINGS
*-------------------------------------------------------------------------------------------------------------

* Variables
local vars "gdp_r_adj cpi_sil log_R_esp_er_hp itsN wage_sil" 
local lava ""Real output" "Consumer prices" "Lending rate" "Money inflow (% of stock)" "Nominal wage"" 
local yti ""Percent" "Percent" "Percentage pts" "Percentage pts" "Percent""

* irf horizon
local H 5
local F 5

* lags
local L 2
local LS `L'
local P 500

*-------------------------------------------------------------------------------------------------------------

local count = 0
foreach v of local vars {
local count = `count'+1
n display "variable `v'"

*_________ GET DATA ____________
use "`path'data/working files/prepared.dta", clear 


*____ DECLARE TIME SERIES _____
gen id =1
tsset id year, yearly


gen period = 1
local jj = 1
forvalues y = 0(100)300 {
local jj = `jj'+1
	replace period = `jj' if year >= 1500 + `y' 
	}
tab period, gen(ppp)
order year period



*____ GENERATE RANDOM RESHUFFLINGS OF PERCLOSS _____
set seed 1

forvalues p = 0/`P' { //placebo runs
n display "variable `v' - run `p'"

*_________________ LOCAL PROJECTIONS _________________

***_________ CONTROLS

local indep	" gdp_r_adj_gr cpi_sil_gr wage_sil_gr itsN_gr" 

forvalues i=0/`L'{
	local indep_`i' = subinstr("`indep'"," "," l`i'_",.)
	}
	
local rhs ""
forvalues i = 1/`L'{
	local rhs "`rhs'`indep_`i''"
	}
	
	local not l0_`v'_gr l0_its_gr
	local rhs: list rhs- not
	
	


	***_________ IMPULSE RESPONSE variables
	if (`p'==0) {
		gen irf_`v' = . if _n<=`H'+1
		}
	else {
		gen irf_`v'_`p' = . if _n<=`H'+1
		}
		

	// reshuffle shock variable (jointly with salvaging and pirate loss variables)
	shufflevar perclossN percloss_salvN percloss_pirN perclossN_vellon, joint

	tsset id year

	***_________ ESTIMATE COEFFICIENTS
	
	forvalues y = 0/`H' {
		n display "projection `y'"


		if (`p'==0) {
			reg `v'_gr`y' l.`v'_gr l(-`F'/`LS').perclossN_vellon l(-`F'/`LS').perclossN `rhs' l(-`F'/`LS').percloss_salvN l(-`F'/`LS').percloss_pirN l(-`F'/`LS').temperature l(-`F'/`LS').warbeu l(-`F'/`LS').warbciv, vce(robust) 
			}
		else {
			reg `v'_gr`y' l.`v'_gr l(-`F'/`LS').perclossN_vellon_shuffled l(-`F'/`LS').perclossN_shuffled `rhs' l(-`F'/`LS').percloss_salvN_shuffled l(-`F'/`LS').percloss_pirN_shuffled l(-`F'/`LS').temperature l(-`F'/`LS').warbeu l(-`F'/`LS').warbciv, vce(robust) 
			}
		
		* Shock response
		if (`p'==0) {
			lincom perclossN 
			replace irf_`v' = r(estimate) if _n==`y'+1 /* coefficients */
			}
		else {
			lincom perclossN_shuffled 
			replace irf_`v'_`p' = r(estimate) if _n==`y'+1 /* coefficients */
			}
		//restore
		}			


	} //placebo runs
	

// select percentiles
egen ub_`v' = rowpctile(irf_`v'_*), p(15.9)
egen lb_`v' = rowpctile(irf_`v'_*), p(84.1)
egen ub90_`v' = rowpctile(irf_`v'_*), p(90)
egen lb90_`v' = rowpctile(irf_`v'_*), p(10)

	
* plot prerequisites
gen time = _n-1 if(_n<=`H'+1)
drop if time > `H'
drop if time == . 
gen zero = 0 if time!=.

local title : word `count' of `lava'
 local ytitle : word `count' of `yti'
local lsize "medlarge"
local tsize "medlarge"

twoway  (rarea ub90_`v' lb90_`v' time, fcolor(gs12) lcolor(gs12)) ///
		(line irf_`v' time, lcolor(black) lpattern(solid) lwidth(medthick)) ///
		(line zero time, lcolor(black) lwidth(medium)), title("") xtitle("Year", size(`lsize')) ytitle(`ytitle', height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(, format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(off) title(`title', size(`tsize')) ///
		graphregion(color(white)) plotregion(color(white)) name(g`count', replace) nodraw

} // vars

graph combine g4 g1 g3 g2 g5, name(LP_placebo_presi, replace) cols(3) imargin(medsmall) plotregion(color(white)) graphregion(color(white)) 
graph display LP_placebo_presi, ysize(4) xsize(6)
graph export "results/FigureB11.pdf", replace 	
	

}


*_________________________________________

			
